

# NAG Library Routine Document

## F02ECF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F02ECF computes selected eigenvalues and eigenvectors of a real general matrix.

### 2 Specification

```

SUBROUTINE F02ECF (CRIT, N, A, LDA, WL, WU, MEST, M, WR, WI, VR, LDVR,      &
                  VI, LDVI, WORK, LWORK, IWORK, BWORK, IFAIL)
INTEGER          N, LDA, MEST, M, LDVR, LDVI, LWORK, IWORK(N), IFAIL
REAL (KIND=nag_wp) A(LDA,N), WL, WU, WR(N), WI(N), VR(LDVR,MEST),      &
                  VI(LDVI,MEST), WORK(LWORK)
LOGICAL          BWORK(N)
CHARACTER(1)     CRIT

```

### 3 Description

F02ECF computes selected eigenvalues and the corresponding right eigenvectors of a real general matrix  $A$ :

$$Ax_i = \lambda_i x_i.$$

Eigenvalues  $\lambda_i$  may be selected either by *modulus*, satisfying:

$$w_l \leq |\lambda_i| \leq w_u,$$

or by *real part*, satisfying:

$$w_l \leq \operatorname{Re}(\lambda_i) \leq w_u.$$

Note that even though  $A$  is real,  $\lambda_i$  and  $x_i$  may be complex. If  $x_i$  is an eigenvector corresponding to a complex eigenvalue  $\lambda_i$ , then the complex conjugate vector  $\bar{x}_i$  is the eigenvector corresponding to the complex conjugate eigenvalue  $\bar{\lambda}_i$ . The eigenvalues in a complex conjugate pair  $\lambda_i$  and  $\bar{\lambda}_i$  are either both selected or both not selected.

### 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Arguments

1: CRIT – CHARACTER(1) *Input*

*On entry:* indicates the criterion for selecting eigenvalues.

CRIT = 'M'

Eigenvalues are selected according to their moduli:  $w_l \leq |\lambda_i| \leq w_u$ .

CRIT = 'R'

Eigenvalues are selected according to their real parts:  $w_l \leq \operatorname{Re}(\lambda_i) \leq w_u$ .

*Constraint:* CRIT = 'M' or 'R'.

- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: A(LDA,N) – REAL (KIND=nag\_wp) array *Input/Output*  
*On entry:* the  $n$  by  $n$  general matrix  $A$ .  
*On exit:* contains the Hessenberg form of the balanced input matrix  $A'$  (see Section 9).
- 4: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F02ECF is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 5: WL – REAL (KIND=nag\_wp) *Input*  
6: WU – REAL (KIND=nag\_wp) *Input*  
*On entry:*  $w_l$  and  $w_u$ , the lower and upper bounds on the criterion for the selected eigenvalues (see CRIT).  
*Constraint:*  $WU > WL$ .
- 7: MEST – INTEGER *Input*  
*On entry:* the second dimension of the arrays VR and VI as declared in the (sub)program from which F02ECF is called. MEST must be an upper bound on  $m$ , the number of eigenvalues and eigenvectors selected. No eigenvectors are computed if  $MEST < m$ .  
*Constraint:*  $MEST \geq \max(1, m)$ .
- 8: M – INTEGER *Output*  
*On exit:*  $m$ , the number of eigenvalues actually selected.
- 9: WR(N) – REAL (KIND=nag\_wp) array *Output*  
10: WI(N) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the first  $M$  elements of WR and WI hold the real and imaginary parts, respectively, of the selected eigenvalues; elements  $M + 1$  to  $N$  contain the other eigenvalues. Complex conjugate pairs of eigenvalues are stored in consecutive elements of the arrays, with the eigenvalue having positive imaginary part first. See also Section 9.
- 11: VR(LDVR,MEST) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* contains the real parts of the selected eigenvectors, with the  $i$ th column holding the real part of the eigenvector associated with the eigenvalue  $\lambda_i$  (stored in WR( $i$ ) and WI( $i$ )).
- 12: LDVR – INTEGER *Input*  
*On entry:* the first dimension of the array VR as declared in the (sub)program from which F02ECF is called.  
*Constraint:*  $LDVR \geq \max(1, N)$ .
- 13: VI(LDVI,MEST) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* contains the imaginary parts of the selected eigenvectors, with the  $i$ th column holding the imaginary part of the eigenvector associated with the eigenvalue  $\lambda_i$  (stored in WR( $i$ ) and WI( $i$ )).

- 14: LDVI – INTEGER *Input*  
*On entry:* the first dimension of the array VI as declared in the (sub)program from which F02ECF is called.  
*Constraint:*  $LDVI \geq \max(1, N)$ .
- 15: WORK(LWORK) – REAL (KIND=nag\_wp) array *Workspace*  
 16: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F02ECF is called.  
*Constraint:*  $LWORK \geq \max(1, N \times (N + 4))$ .
- 17: IWORK(N) – INTEGER array *Workspace*
- 18: BWORK(N) – LOGICAL array *Workspace*
- 19: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, CRIT  $\neq$  'M' or 'R',  
 or  $N < 0$ ,  
 or  $LDA < \max(1, N)$ ,  
 or  $WU \leq WL$ ,  
 or  $MEST < 1$ ,  
 or  $LDVR < \max(1, N)$ ,  
 or  $LDVI < \max(1, N)$ ,  
 or  $LWORK < \max(1, N \times (N + 4))$ .

IFAIL = 2

The *QR* algorithm failed to compute all the eigenvalues. No eigenvectors have been computed.

IFAIL = 3

There are more than MEST eigenvalues in the specified range. The actual number of eigenvalues in the range is returned in M. No eigenvectors have been computed. Rerun with the second dimension of VR and  $VI = MEST \geq M$ .

IFAIL = 4

Inverse iteration failed to compute all the specified eigenvectors. If an eigenvector failed to converge, the corresponding column of VR and VI is set to zero.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

If  $\lambda_i$  is an exact eigenvalue, and  $\tilde{\lambda}_i$  is the corresponding computed value, then

$$|\tilde{\lambda}_i - \lambda_i| \leq \frac{c(n)\epsilon\|A'\|_2}{s_i},$$

where  $c(n)$  is a modestly increasing function of  $n$ ,  $\epsilon$  is the *machine precision*, and  $s_i$  is the reciprocal condition number of  $\lambda_i$ ;  $A'$  is the balanced form of the original matrix  $A$  (see Section 9), and  $\|A'\| \leq \|A\|$ .

If  $x_i$  is the corresponding exact eigenvector, and  $\tilde{x}_i$  is the corresponding computed eigenvector, then the angle  $\theta(\tilde{x}_i, x_i)$  between them is bounded as follows:

$$\theta(\tilde{x}_i, x_i) \leq \frac{c(n)\epsilon\|A'\|_2}{sep_i},$$

where  $sep_i$  is the reciprocal condition number of  $x_i$ .

The condition numbers  $s_i$  and  $sep_i$  may be computed from the Hessenberg form of the balanced matrix  $A'$  which is returned in the array A. This requires calling F08PEF (DHSEQR) with JOB = 'S' to compute the Schur form of  $A'$ , followed by F08QLF (DTRSNA).

## 8 Parallelism and Performance

F02ECF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F02ECF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

F02ECF calls routines from LAPACK in Chapter F08. It first balances the matrix, using a diagonal similarity transformation to reduce its norm; and then reduces the balanced matrix  $A'$  to upper Hessenberg form  $H$ , using an orthogonal similarity transformation:  $A' = QHQ^T$ . The routine uses the Hessenberg  $QR$  algorithm to compute all the eigenvalues of  $H$ , which are the same as the eigenvalues

of  $A$ . It computes the eigenvectors of  $H$  which correspond to the selected eigenvalues, using inverse iteration. It premultiplies the eigenvectors by  $Q$  to form the eigenvectors of  $A'$ ; and finally transforms the eigenvectors to those of the original matrix  $A$ .

Each eigenvector  $x$  (real or complex) is normalized so that  $\|x\|_2 = 1$ , and the element of largest absolute value is real and positive.

The inverse iteration routine may make a small perturbation to the real parts of close eigenvalues, and this may shift their moduli just outside the specified bounds. If you are relying on eigenvalues being within the bounds, you should test them on return from F02ECF.

The time taken by the routine is approximately proportional to  $n^3$ .

The routine can be used to compute *all* eigenvalues and eigenvectors, by setting WL large and negative, and WU large and positive.

## 10 Example

This example computes those eigenvalues of the matrix  $A$  whose moduli lie in the range  $[0.2, 0.5]$ , and their corresponding eigenvectors, where

$$A = \begin{pmatrix} 0.35 & 0.45 & -0.14 & -0.17 \\ 0.09 & 0.07 & -0.54 & 0.35 \\ -0.44 & -0.33 & -0.03 & 0.17 \\ 0.25 & -0.32 & -0.13 & 0.11 \end{pmatrix}.$$

### 10.1 Program Text

```

Program f02ecfe

!      F02ECF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: f02ecf, nag_wp, x04dbf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)         :: wl, wu
      Integer                    :: i, ifail, j, lda, ldv, ldvi, ldvr, &
                                lwork, m, mest, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: v(:, :)
      Real (Kind=nag_wp), Allocatable   :: a(:, :), vi(:, :), vr(:, :), wi(:, &
                                work(:), wr(:)
      Integer, Allocatable              :: iwork(:)
      Logical, Allocatable              :: bwork(:)
      Character (1)                    :: clabs(1), rlabs(1)
!      .. Intrinsic Procedures ..
      Intrinsic                       :: cplx
!      .. Executable Statements ..
      Write (nout,*) 'F02ECF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, mest, wl, wu
      lda = n
      ldv = n
      ldvi = n
      ldvr = n
      lwork = 64*n
      Allocate (v(ldv,n),a(lda,n),vi(ldvi,mest),vr(ldvr,mest),wi(n), &
                work(lwork),wr(n),iwork(n),bwork(n))

!      Read A from data file

```

```

      Read (nin,*)(a(i,1:n),i=1,n)

!      Compute selected eigenvalues and eigenvectors of A

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call f02ecf('Moduli',n,a,lda,wl,wu,mest,m,wr,wi,vr,ldvr,vi,ldvi,work, &
        lwork,iwork,bwork,ifail)

      Write (nout,*)
      Write (nout,*) 'Eigenvalues'
      Write (nout,99999)(' (',wr(i),',',wi(i),')',i=1,m)
      Write (nout,*)
      Flush (nout)
      Do j = 1, m
        Do i = 1, n
          v(i,j) = cplx(vr(i,j),vi(i,j),kind=nag_wp)
        End Do
      End Do

      ifail = 0
      Call x04dbf('General',' ',n,m,v,ldv,'Bracketed','F7.4','Eigenvectors', &
        'Integer',rlabs,'Integer',clabs,80,0,ifail)

99999 Format ((3X,4(A,F7.4,A,F7.4,A,:))
      End Program f02ecfe

```

## 10.2 Program Data

```

F02ECF Example Program Data
  4 3 0.2 0.5          : n, mest wl, wu
  0.35  0.45 -0.14 -0.17
  0.09  0.07 -0.54  0.35
 -0.44 -0.33 -0.03  0.17
  0.25 -0.32 -0.13  0.11 : matrix A

```

## 10.3 Program Results

F02ECF Example Program Results

Eigenvalues  
 (-0.0994, 0.4008) (-0.0994,-0.4008)

Eigenvectors

	1	2
1	(-0.1933, 0.2546)	(-0.1933,-0.2546)
2	( 0.2519,-0.5224)	( 0.2519, 0.5224)
3	( 0.0972,-0.3084)	( 0.0972, 0.3084)
4	( 0.6760, 0.0000)	( 0.6760,-0.0000)

---